Journal of Integer Sequences, Vol. 28 (2025),
Article 25.3.8

Using `Walnut` to Solve Problems from the OEIS

Wieb Bosma
Department of Mathematics
Radboud University
Heyendaalseweg 135, 6525 AJ Nijmegen
The Netherlands
bosma@math.ru.nl

René Bruin
Mathematisch Instituut
Utrecht University
Budapestlaan 6, 3584 CD Utrecht
The Netherlands
r.bruin@uu.nl

Robbert Fokkink*
Faculty of Mathematics
Technical University Delft
Mekelweg 4, 2628 CD Delft
The Netherlands
r.j.fokkink@tudelft.nl

Jonathan Grube
Mathematisch Instituut
Utrecht University
Budapestlaan 6, 3584 CD Utrecht
The Netherlands
j.f.p.grube@students.uu.nl

Anniek Reuijl
Mathematisch Instituut
Utrecht University
Budapestlaan 6, 3584 CD Utrecht
The Netherlands
a.reuijl@students.uu.nl

Thian Tromp
Mathematisch Instituut
Utrecht University
Budapestlaan 6, 3584 CD Utrecht
The Netherlands
t.a.n.tromp@students.uu.nl

Abstract

We use the automatic theorem prover `Walnut` to resolve various open problems from the OEIS (On-Line Encyclopedia of Integer Sequences) and beyond. Specifically, we clarify the structure of sequence A260311, which concerns runs of sums of upper Wythoff numbers. We extend a result of Hajdu, Tijdeman, and Varga on polynomials with nonzero coefficients modulo a prime. Additionally, we settle open problems related to the anti-recurrence sequences A265389 and A299409, as well as the sumfree sequences A026471 and A026475. Our findings also give rise to new open problems.

*Robbert Fokkink is the corresponding author.

1 Introduction

In the fall of 2024 two of us (Bosma and Fokkink) taught a course on automatic sequences in the Dutch national mathematics master's programs, based on Shallit's monograph [14]. The aim of the course was to allow students to solve as many problems as possible from the On-Line Encyclopedia of Integer Sequences (OEIS) using the automatic theorem prover `Walnut` [10]. This paper contains our results. It consists of four distinct sections: an analysis of sums of upper Wythoff numbers, an investigation of polynomials with rational roots, and two separate studies on specific sumfree sequences. All these studies are carried out with `Walnut`.

`Walnut` is a free software package, which can verify first-order logic statements on sets or sequences of natural numbers involving addition but not multiplication. Statements can be expressed in various numeration systems, such as binary, decimal, or the more exotic Zeckendorf numeration system, in which numbers are represented by sums of Fibonacci numbers. It can be downloaded from https://cs.uwaterloo.ca/~shallit/walnut.html.

The syntax of `Walnut` is highly transparent and can be easily understood by anyone familiar with first-order logic. To illustrate this, here is an example. The Fibonacci word $\mathbf{F} = abaababaabaab\cdots$ is the infinite symbolic sequence generated by the substitutions $a \mapsto ab$ and $b \mapsto a$. A well-known property of $\mathbf{F}$ is that each prefix of length $F_n - 2$ is a palindrome [4], where F_n is the n-th Fibonacci number. This prefix runs up to the index $F_n - 3$ if we start the count at zero. In first-order logic, the palindromic property can be expressed as

$$\forall F_n \geq 3\ \forall j \leq F_n - 3\colon\ \mathbf{F}[j] = \mathbf{F}[F_n - 3 - j].$$

The Fibonacci numbers $0, 1, 1, 2\ldots$ are denoted $F_0, F_1, F_2, F_3, \ldots$. In particular, F_4 is the first Fibonacci number such that $F_n \geq 3$. A `Walnut` verification of the palindromic property is

```
eval prefixtest "?msd_fib Af Aj ($isfib(f)&f>=3&j<=f-3) => F[j]=F[f-3-j]":
```

`Walnut` evaluates the hyphenated statement, which is named `prefixtest`. The command `?msd_fib` says that the statement should be evaluated in the Zeckendorf system. The acronym `msd` stands for most significant digit first. Capital A represents $\forall$, and similarly, E represents $\exists$. The command `$isfib(f)` checks whether `f` is a Fibonacci number. The Fibonacci word $\mathbf{F}$ is implemented in `Walnut` as `F`. The command should now be clear. It is evaluated as `TRUE`.

2 Gaps between unsums of upper Wythoff numbers

The upper Wythoff numbers are the indices for the letter b in the word $\mathbf{F}$, if we start the count at one instead of zero. The first few upper Wythoff numbers are

$$2,\ 5,\ 7,\ 10,\ 13,\ 15,\ 18,\ 20,\ 23,\ 26,\ 28,\ 31,\ \ldots\ .$$



Figure 1: An automaton that accepts sequence A260317 in Zeckendorf numeration. Accepting states are marked by double circles. For instance, 101 which represents 4 in Zeckendorf numeration is accepted, and 1010, which represents 7, is rejected.

Kawsumarng, Khemaratchatakumthorn, Noppakaew, and Pongsriiam [8] studied, among other things, the set $U+U$ of sums of upper Wythoff numbers, consisting of sums $u+v$ with $u, v \in U$. Shallit [15] derived their results using `Walnut`. A subsequent study [12] extended the work to sums of Beatty sequences.

Let $U \oplus U$ be the set of sums $u+v$ such that $u \neq v$. Sequence A260317 in the OEIS is the complement of $U \oplus U$. We say that the numbers in A260317 are *unsums*. For instance, $4 = 2+2$ is an unsum even though it is the sum of two upper Wythoff numbers, since these numbers are not distinct. The first few unsums are

$$1,\ 2,\ 3,\ 4,\ 5,\ 6,\ 8,\ 10,\ 11,\ 13,\ 14,\ 16,\ 19,\ \ldots\ .$$

Sequence A260311 in the OEIS contains the differences between consecutive unsums. We say that they are *gaps*. The initial gaps are

$$1111122121232323323323532353235353235353235535535323535535 32.$$

Shallit [16] implemented A260317 in `Walnut` to prove that the gaps are limited to 1, 2, 3, and 5. This resolved a problem posed by Kimberling. The automaton that recognizes the unsums is illustrated in Fig. 1. Building on this work, we extend the analysis of A260311 and fully clarify its structure.

The prefix 1111122121 is irregular, but after that a pattern emerges. We first describe what we see and then we check that with `Walnut`. The words between the gaps 2 are palindromes of increasing lengths:

$$2W_1 2W_1 2W_2 2W_2 2W_3 2W_3 2 \cdots\ .$$

The gaps 2 are markers around the palindromes. Each palindrome W_i occurs twice and has Fibonacci length $|W_i| = F_{i+1}$. The sequence of palindromes $W_1, W_2, W_3, \ldots$ is

$$3,\ 33,\ 353,\ 35353,\ 35355353,\ 3535535355353,\ 353553535535535355353,\ \ldots$$

Each W_i is of the form $3V_i3$ for $i > 1$ and $3V_i$ is a prefix of W_{i+1}. For odd i we even find that W_i is a prefix of W_{i+1}. The sum of the digits of the palindromes is equal to

$$3,\ 6,\ 11,\ 19,\ 32,\ 53,\ 87,\ \ldots.$$

If we add 2 to these sums, which is the value of the markers, then we get the Fibonacci numbers $F_5, F_6, F_7, F_8 \ldots$. The limit of the sequence of palindromes is the infinite word

$$353553553553553553535535535535355355\cdots.$$

If we code the Fibonacci word $\mathbf{F}$ by $a \to 5$ and $b \to 3$, then this limit is $3\mathbf{F}$. We are going to check all these observed properties with `Walnut`.

Lemma 1. *The tenth gap is the last gap that is equal to* 1.

Proof. We need to check that the final occurrence of 1 is at the end of the prefix 1111122121. The digit sum of this prefix is 13 and the first entry of A260311 is 1. That is why the final entry 1 in our prefix is the gap between the unsums 13 and 14. We verify that this is indeed the final occurrence of that gap:

```
eval test "?msd_fib An ($a260317(n) & $a260317(n+1)) => n<14":
```

`Walnut` returns `TRUE`. □

The digits 2 are markers and we single them out. Since the final gap 1 occurs at 13, all gaps are ≥ 2 for $n > 13$:

```
def marker2 "?msd_fib $a260317(n) & $a260317(n+2) & n>13":
```

The command `marker2` accepts unsums > 13 that have a gap of size 2.

Lemma 2. *Let W be a word in between two consecutive markers 2. The sum of the digits of W plus two is a Fibonacci number.*

Proof. Since the digits in W are gaps, their sum plus two is a difference between consecutive markers. We collect the differences in `gapmark2` and check that these are the Fibonacci numbers that are greater than 4.

```
def gapmark2 "?msd_fib En (t>0) & $marker2(n) & $marker2(n+t) &
(As (s<t & s>0) => ~$marker2(n+s))":
eval test "?msd_fib An $gapmark2(n) <=> (n>4 & $isfib(n))":
```

`Walnut` returns `TRUE`. □

F_5 is the first Fibonacci number that is greater than 4. So, all digit sums plus two are F_n for $n \geq 5$. It is a bit tiresome to say digit sum plus two all the time. Therefore, we speak of the sum of the palindrome, even though we have not verified yet that the words between markers are indeed palindromes.

Lemma 3. *The sums of the palindromes form the sequence* $F_5, F_5, F_6, \ldots$. *Each Fibonacci number is repeated twice, starting from* F_5.

Proof. A sum of Fibonacci numbers is Fibonacci if and only if the numbers are consecutive: $F_k + F_m = F_n$ if and only if $m = k + 1$ and $n = m + 1$. The command `triplegap` accepts t, u, v that are sums of three consecutive palindromes.

```
def triplegap "?msd_fib En (t>0) & (u>0) & (v>0) & $marker2(n) &
$marker2(n+t) & $marker2(n+t+u) & $marker2(n+t+u+v) &
(Aw (w>0) & (w<t+u+v) & (w!=t&w!=t+u) => ~$marker2(n+w))":
```

We now check that for such triples t, u, v the sum $t + v$ is a Fibonacci number.

```
eval test "?msd_fib At,u,v $triplegap(t,u,v) <=>
( $isfib(t) & $isfib(u) & $isfib(v) & $isfib(t+v) &
(t<=u) & (u<=v) & (t>4) )":
```

`Walnut` says `TRUE` in both cases. Each triple t, u, v must be of the form (F_n, F_n, F_{n+1}) or (F_n, F_{n+1}, F_{n+1}). The sequence starts out with (F_5, F_5, F_6) and continues as described. □

We proved that sums of palindromes are repeated. We still need to show that the palindromes are repeated, but we already label the sequence as W_i in anticipation of the fact that they form the sequence $W_1, W_1, W_2, \ldots$. First, we prove that these words are indeed palindromes.

Lemma 4. *Each* W_i *is a palindrome.*

Proof. Suppose that W is marked, i.e., $2W2$, and that the markers correspond to the unsums $m < n$. Then the digits in W correspond to the gaps between the unsums $m + 2 < k < n$. W is a palindrome if and only if the reversal $x \mapsto n + m + 2 - x$ of the interval $[m + 2, n]$ preserves the subset of unsums.

```
eval test "?msd_fib Ak,m,n ( m>14 & k>m+2 & n>k )
(& $marker2(m) & $marker2(n) & (Ap((p<n) & (p>m)) => ~$marker2(p))) => (
$a260317(k) <=> $a260317(n+m+2-k) )":
```

`Walnut` says `TRUE`. □

Lemma 5. *Each* W_i *starts and ends with digit* 3.

Proof. Since the word is a palindrome, we only need to prove that the initial digit is 3. In other words, every marker is followed by gap 3. Equivalently, if n and $n + 2$ are unsums for $n \geq 14$, then $n + 5$ is an unsum.

```
eval test "?msd_fib A n (n>=14 & $marker2(n)) => $a260317(n+5)":
```

`Walnut` says `TRUE`. □

We will now prove that each palindrome is repeated twice. We can write $W = 3V3$, except for the first palindrome which is the single letter 3.

Lemma 6. *Let $W = 3V3$ be a palindrome. Then $3V$ is the prefix of the next palindrome. In particular, the next palindrome is the same if its sum is the same.*

Proof. Let m and n be unsums for consecutive markers. We need to verify that the gaps that follow m are equal to the gaps that follow n, up to the next marker.

```
eval test "?msd_fib A m,n (n>m & $marker2(m) & $marker2(n) & (Ap ((p<n) &
(p>m)) => ~$marker2(p))) => (Ak k>m+2 & k<n => ($a260317(k)<=>
$a260317(n-m+k)))":
```

`Walnut` says `TRUE`. The next palindrome has prefix $3V$. If its sum is the same, then it must be the same palindrome $3V3$. □

We conclude that the palindromes are repeated twice $W_1, W_1, W_2, \ldots$.

Lemma 7. *W_i is a prefix of W_{i+1} if and only if i is odd.*

Proof. The sum of W_i is equal to F_{i+4}. The parity of the indices is the same. We can check the parity of the index of a Fibonacci number with `Walnut`.

```
reg isoddfib msd_fib "0*1(00)*0":
```

If we add the parity check to the `Walnut` code for the proof of Lemma 4, then we can extend the prefix by one letter.

```
eval test "?msd_fib A m,n (n>m & $marker2(m) & $marker2(n) & $isoddfib(n-m)
& (Ap ((p<n) & (p>m)) => ~$marker2(p))) => (Ak k>m+2 & k<=n =>
($a260317(k)<=> $a260317(n-m+k)))":
```

`Walnut` says `TRUE`. This proves that W_i is a prefix of W_{i+1} if i is odd. Note that we just copy-pasted the code from the proof of Lemma 6 and included one more gap by changing `k<n` to `k<=n`. We could have also simply checked `$a260317(2*n-m)`.

We turn to W_i for even i. We prove that it is not a prefix of W_{i+1} and now we need to make sure that the next palindrome is indeed different. Therefore, we need to include the condition that the next marker is different from $2n - m$. It suffices to check `~$a260317(2*n-m)`.

```
eval test "?msd_fib A m,n (n>m & $marker2(m) & $marker2(n) &
~$marker2(2*n-m) & $isevenfib(n-m) & (Ap ((p<n) & (p>m)) => ~$marker2(p)))
=> ~$a260317(2*n-m)":
```

`Walnut` says `TRUE`. □

Let $W_i = 3V_i3$ for $i \geq 2$ (remember $W_1 = 3$ and $W_2 = 33$).

Lemma 8. *If $i \geq 2$ is odd then $W_{i+1} = W_i 5 V_{i-1} 3$ and if it is even then $W_{i+1} = 3 V_i 5 W_{i-1}$. In particular, $|W_{i+1}| = |W_i| + |W_{i-1}|$ with $|W_1| = 1$ and $|W_2| = 2$. The lengths of the palindromes are the Fibonacci numbers.*

Proof. By Lemma 7, W_{i+1} has prefix $3V_i$. It has suffix $V_i 3$ since both W_{i+1} and V_i are palindromes. It follows that W_{i+1} has suffix $V_{i-1}3$. By computing sums of the digits, we find that the prefix $3V_i$ and the suffix $V_{i-1}3$ do not overlap. The sum of the digits of W_{i+1} is $F_{i+5} - 2$. The sum of the digits of the prefix $3V_i$ and the suffix $V_{i-1}3$ is

$$F_{i+4} - 5 + F_{i+3} - 5 = F_{i+5} - 10.$$

We still lack a digit 3 and a digit 5. Either $W_{i+1} = 3V_i 35 V_{i-1}3$ or $W_{i+1} = 3V_i 53 V_{i-1}3$. In the first case i is odd and in the second case i is even. □

Lemma 9. *Let $\mathbf{V}$ be the image of the Fibonacci word $\mathbf{F}$ under the coding $a \mapsto 5$ and $b \mapsto 3$. The limit of W_i equals $3\mathbf{V}$.*

Proof. We need to prove that V_i converges to $\mathbf{V}$. We have that

$$V_{i+1} = \begin{cases} V_i 35 V_{i-1} & \text{if } i \text{ is odd,} \\ V_i 53 V_{i-1} & \text{if } i \text{ is even.} \end{cases}$$

If we put $P_i = V_i 35$ if i is odd and $P_i = V_i 53$ if i is even, then $P_{i+1} = P_i P_{i-1}$ starting from $P_2 = 53$ and $P_3 = 535$. This is a well-known recursion that generates the Fibonacci word, see [4]. The palindromes of length $F_i - 2$ that we encountered in the introduction are the V_i. Since P_i has the same limit as V_i, we are done. □

This concludes our verification of the observed properties.

3 Polynomials without zero coefficients

Fine [6] proved that if all the coefficients of the polynomial $(x+1)^n$ are odd, then $n = 2^m - 1$ for some m. We can verify that with `Walnut`. These coefficients are the binomials $\binom{n}{k}$ for $0 \leq k \leq n$. Fine's result depends on a classical theorem of Lucas, according to which for integers k, n, and prime p, the following holds:

$$\binom{n}{k} \equiv \prod_{j=0}^{r} \binom{n_i}{k_i} \bmod p,$$

where k_i, n_i are the digits of the base p expansions of $k = k_r p^r + \cdots + k_1 p + k_0$ and $n = n_r p^r + \cdots + n_1 p + n_0$, and we use the convention that $\binom{a}{b} = 0$ if $a < b$. In particular, $\binom{n}{k}$ is odd if and only if there does not exist an i such that $k_i = 1$ and $n_i = 0$. In other words, all digits (k_i, n_i) are in $\{(0,0), (0,1), (1,1)\}$. Translating this into `Walnut` syntax gives

```
reg bincoef msd_2 msd_2 "([0,0]|[0,1]|[1,1])*":
```

The binary numeration system is given by `msd_2`. A power of 2 in binary is a 1 followed by a string of 0's:

```
reg power2 msd_2 "0*10*":
```

Fine's theorem in first-order logic is

$$\forall n \ (\exists m \ n+1 = 2^m) \Leftrightarrow \left(k \leq n \implies \binom{n}{k} \equiv 1 \bmod 2\right).$$

We can now check that with `Walnut`, where we write n as j to adhere to the convention that variables are entered in alphabetical order:

```
eval fine "Aj $power2(j+1) <=> (Ak (k<=j => $bincoef(k,j)))":
```

which evaluates as `TRUE`.

In a recent paper Hajdu et al. [7] extended Fine's theorem to polynomials modulo three.

Theorem 10 (Hajdu et al.)**.** *If none of the coefficients of $(x-1)^c(x+1)^d$ is divisible by 3, then $c+d+1$ is of the shape $3^j, 2\cdot 3^j, 3^i+3^j, 2\cdot 3^i+3^j$ for $i>j\geq 0$.*

The question we would like to begin to address here can be formulated as follows: what can be said about the degree $\deg f$ of polynomials f over a finite field $\mathbb{F}_q$ of q elements for which all irreducible factors are linear without zero coefficients?

The theorems of Fine and Hajdu et al. show that for $q=2$ and $q=3$ it holds that the number of non-zero digits in the binary, resp., ternary expansion of $(\deg f)+1$ is bounded, by 1 and 2 respectively.

As a special case of a deep result by Adamczewski et al. [2] on formal power series, it is possible to find an automaton with output that will help us answer such a question. This automaton `co`q, will consist of q^{q-1} states. The purpose of this automaton is that, upon input a q-tuple $(k, a_1, a_2, \ldots, a_{q-1})$, it outputs the k-th coefficient $\mathrm{co}_q(k, a_1, ..., a_{q-1})$ of the polynomial

$$f = (x-\alpha_1)^{a_1}(x-\alpha_2)^{a_2}\cdots(x-\alpha_{q-1})^{a_{q-1}} \in \mathbb{F}_q[x],$$

where the α_i are the distinct elements of $\mathbb{F}_q$. The existence of such an automaton follows from the observation that, summing over all tuples $0 \leq k, a_1, ..., a_{q-1} < \infty$,

$$\sum \mathrm{co}_q(k, a_1, ..., a_{q-1}) x^k y_1^{a_1}\cdots y_{q-1}^{a_{q-1}} = \prod_{i=1}^{q} \frac{1}{(1-y_i(x-\alpha_i))},$$

(using the familiar geometric series identity repeatedly), which is a multivariate power series that is *rational* over $\mathbb{F}_q(x, y_1, \ldots, y_{q-1})$. Automaticity then results from the multivariate generalization of Christol's theorem by Adamczewski and Bell [1], while [2] contains an efficient algorithm to compute the automaton. Using an implementation of this algorithm in



Figure 2: The `lsd_3`-automaton that accepts $n = c+d+1$ such that none of the coefficients in $(X-1)^c(X+1)^d$ are divisible by three. If a transition is missing, the input is rejected. For instance, 21 is represented by 012 and is accepted, while 22 is represented by 112 and is rejected.

Magma [3], we generated the automata for $q = 3, 4$, and 5. They become large very quickly, since one has to specify the q^q possible transitions for each of the q^{q-1} states: leading to a specification for a text file in `Walnut` with almost 2 million lines for $p = 5$.

Starting with the modest automaton `co3` having 9 states and 243 transitions, one computes any coefficient of $(x-1)^c(x+1)^d$ modulo three from three `lsd_3` integers k, c, d as input.

To verify its correctness, the effect of multiplication by $x-1$ and $x+1$ for each state can be checked. This gives eighteen verifications in total; we list three of them (as predicates without the `eval` command, to save space):

```
"?lsd_3 Aa,c,d(co3[a+1][c][d]=@0&co3[a][c][d]=@0)=>co3[a+1][c+1][d]=@0":
"?lsd_3 Aa,c,d(co3[a+1][c][d]=@1&co3[a][c][d]=@0)=>co3[a+1][c+1][d]=@2":
"?lsd_3 Aa,c,d(co3[a+1][c][d]=@2&co3[a][c][d]=@0)=>co3[a+1][c+1][d]=@1":
```

These check, for each of the three possible values, that coefficient $a+1$ of $(x-1)\cdot f$ is obtained as the difference of coefficients a and $a+1$ of the original f.

With `co3` it is possible in `Walnut` to check that Theorem 10 holds, as follows.

```
eval co3no0 "?lsd_3 Ec,d c+d+1=n&(Aa a<=n-1=>co3[a][c][d]!=@0)":
```

The resulting automaton `co3no0` for `n` is shown in Fig. 2. This does indeed confirm Theorem 10, as the largest number of non-zero digits in an input path (for n) leading to an accepting state is equal to 3.

Hajdu et al. ask if a far-reaching generalization of these results holds: is it true that for every prime p there exists a constant c_p such that any monic polynomial $f(x) \in \mathbb{Z}[x]$ with only rational roots and no coefficients divisible by p has at most c_p nonzero digits in the base p expansion of $\deg(f)+1$? In particular, is this true for $c_p = p-1$? Much more will be said

about this elsewhere, but using `Walnut` we can prove two results in addition to the cases for $p = 2$ and $p = 3$ above.

The automaton `co5` we produced along the lines sketched above computes the coefficients of $(X-2)^b(X-1)^c(X+1)^d(X+2)^e$ modulo five, but our `Walnut` computation for the degrees of polynomials without coefficients divisible by five spirals out of control even on a powerful machine. As before, for `co3`, the correctness of `co5` could inductively be checked easily (using 5^3 checks in all). It was possible to verify the conjecture for special cases (essentially leaving out some factors in the product), but it was felt that the general case for $p = 5$ also ought to be doable.

Here Nicol and Frohme came to the rescue: they were working on improvements for `Walnut` (version 7), to be reported on in a forthcoming paper [11]. This version did succeed in producing the analogon `co5no0` of `co3no0`. The result is shown in Fig. 3. Again, confirmation



Figure 3: The `msd_5`-automaton that accepts $n = b + c + d + e + 1$ such that none of the coefficients in $(X-2)^b(X-1)^c(X+1)^d(X+2)^e$ is divisible by five. Accepting states are marked by double circles.

of the claims lies in the fact that no path to an accepting state contains more than 4 non-zero symbols.

The conjecture of Hajdu refers to arithmetic modulo primes p, so in a prime field only. We decided to also consider the case of $\mathbb{F}_4$, the field of 4 elements $\{0, 1, g, g^2\}$ where $g^2 = g+1$ over $\mathbb{F}_2$. With some care our algorithm also produced the automaton `co4`, which verifies the k-th coefficient of

$$f = (x+1)^b(x+g)^c(x+g^2)^d \in \mathbb{F}_4[x].$$



Figure 4: The `lsd_4`-automaton that accepts $n = b + c + d + 1$ such that none of the coefficients of product of linear factors of degree $n - 1$ is zero. Accepting states are marked by double circles.

After checking correctness, the automaton `co4` easily produced `co4no0`, shown in Fig. 4. Here we see that 4 is the largest number of non-zero digits in the 4-ary expansion of $(\deg f)+1$. As an example, the path $1, 1, 2, 1$ in the automaton, corresponding to $n = 1+4+2\cdot 4^2+4^3 = 101$ suggests that there exist polynomials over $\mathbb{F}_4$ of degree 100 with no non-zero coefficients that are a product of linear factors. Indeed,

$$f = (x + 1)^{28}(x + g)^{35}(x + g^2)^{37}$$

is readily exhibited by Magma as an example. An automaton recognizing these polynomials is implicitly also constructed by `Walnut`.

4 Anti-recurrence sequences

Don't be unprepared for double negations! Fibonaccis are generated by adding sums of consecutive numbers starting from 1 and 2. In contrast, *anti-Fibonaccis* are generated by deleting sums of consecutive numbers, starting from 1 and 2. Their sum is the anti-Fibonacci 3. We delete this number and move on to the next two consecutive numbers 4 and 5 to get the anti-Fibonacci 9. After that we delete $6 + 7$ and $8 + 10$, and so on. This gives us sequence A075326, in which 0 is added as the first anti-Fibonacci:

$$0,\ 3,\ 9,\ 13,\ 18,\ 23,\ 29,\ 33,\ 39,\ 43,\ 49,\ 53,\ \ldots.$$

The numbers that are missing from this sequence are known as the *non-anti-Fibonaccis*. They can be found in A249031:

$$1,\ 2,\ 4,\ 5,\ 6,\ 7,\ 8,\ 10,\ 11,\ 12,\ 14,\ 15,\ 16,\ \ldots.$$

The anti-Fibonacci numbers were introduced by Hofstadter in an unpublished note. Zaslavsky completely cleared up their structure in another unpublished note. Kimberling and Moses [9] studied general anti-recurrence sequences.

Zaslavsky proved that the gaps between anti-Fibonaccis come in pairs $\{6, 4\}$ and $\{5, 5\}$. If these pairs are labelled a and b respectively, then they form the fixed point of the period-doubling morphism $a \mapsto ab,\ b \mapsto aa$. This sequence is implemented as `PD` in `Walnut` with a represented by zero and b represented by one.

Theorem 11 (Zaslavsky). *The sequence X_n of anti-Fibonacci numbers is given by:*

$$\begin{aligned} X_{2k+1} &= 3 + 10k, \\ X_{2k+2} &= 8 + 10k + \texttt{PD}[k]. \end{aligned}$$

The non-anti-Fibonaccis Y_n are given by:

$$\begin{aligned} Y_{4k} &= 5k, \\ Y_{4k+1} &= 5k + 1, \\ Y_{4k+2} &= 5k + 2, \\ Y_{4k+3} &= 5k + 3 + \texttt{PD}[k]. \end{aligned}$$

Proof. To prove this result in `Walnut`, we implement X_n and Y_n as synchronized sequences. The command `antifib` specifies X_n and `nonafib` specifies Y_n.

```
def antifib "Ek n=2*k+1 & x=3+10*k | Ek n=2*k+2 & PD[k]=@1 & x=9+10*k
| Ek n=2*k+2 & PD[k]=@0 & x=8+10*k":
def nonafib "Ek n=4*k & y=5*k | Ek n=4*k+1 & y=5*k+1 | Ek n=4*k+2 & y=5*k+2
| Ek n=4*k+3 & PD[k]=@1 & y=5*k+4 | Ek n=4*k+3 & PD[k]=@0 & y=5*k+3":
```

Now we need to verify that these sequences do indeed have the required properties: they are complementary and one contains sums of consecutive numbers of the other. In particular $X_n = Y_{2n-1} + Y_{2n}$. We first verify the additive relation between the two sequences:

```
eval zaslavsky1 "An,x,y,z
((n>0) & $nonafib(2*n-1,x) & $nonafib(2*n,y) & $antifib(n,z)) => x+y=z":
```

`Walnut` says `TRUE`. We now verify that the sequences are complementary in two steps. Disjointness is checked by

```
eval zaslavsky2 "Ei,j,n (n>0) & $nonafib(i,n) & $antifib(j,n)":
```

on which `Walnut` returns `FALSE`. We verify that each number is in one of the two sequences

```
eval zaslavsky3 "An (n>0) => (Ej $nonafib(j,n) | $antifib(j,n))":
```

on which `Walnut` returns `TRUE`. The sequences are indeed complementary, establishing that the X_n are the anti-Fibonaccis and the Y_n are the non-anti-Fibonaccis. □

Kimberling and Moses [9] observed that there is another way to describe the non-anti-Fibonaccis Y_n using the `mex` or minimal excluded value. For any set of natural numbers the `mex` is the minimal element of its complement. Create sequences A_n, B_n, C_n starting from $A_1 = 1, B_1 = 2, C_1 = 3$ and iteratively add new numbers $A_{n+1} = \texttt{mex}(\{A_i, B_i, C_i \colon i \leq n\})$

and $B_{n+1} = \texttt{mex}(\{A_i, B_i, C_i : i \le n\} \cup \{A_{n+1}\})$ and $C_{n+1} = A_{n+1} + B_{n+1}$. The C_n are the anti-Fibonaccis. The odd-indexed non-anti-Fibonaccis are A_n and the even-indexed are B_n.

This definition allows us to move beyond anti-Fibonaccis. If we apply the same process to four sequences A_n, B_n, C_n, D_n instead of three, then we get the *anti-Tribonacci* sequence. As in the previous case, A_n, B_n, C_n are defined by the `mex` and D_n is the sum. The D_n are the anti-Tribonaccis A265389:

$$6,\ 16,\ 27,\ 36,\ 46,\ 57,\ 66,\ 75,\ 87,\ 96,\ \ldots\ .$$

The other three sequences are A297464, A297465, A297466. Clark Kimberling conjectured that

$$\begin{array}{rcll} 0 & \le & 10n - 6 - 3A_n & \le 2, \\ 0 & \le & 10n - 2 - 3B_n & \le 3, \\ 0 & \le & 10n + 1 - 3C_n & \le 3, \\ 0 & \le & 10n - 3 - D_n & \le 2. \end{array} \tag{1}$$

Following Zaslavsky's lead, we guess that these four equations form automatic sequences and verify this using `Walnut`. We call them the *remainders* and our guessed 3-automata are given in Figure 5. We have implemented these as `lsd_3` automata `xkimber`, `ykimber`,



Figure 5: The guessed `lsd_3`-automata for the remainders associated to the anti-Tribonacci sequence.

`zkimber`, `wkimber`. The sequences A, B, C, D are given by

```
def seqa "?lsd_3 s=(10*n-6-xkimber[n-1])/3":
def seqb "?lsd_3 s=(10*n-2-ykimber[n-1])/3":
def seqc "?lsd_3 s=(10*n+1-zkimber[n-1])/3":
def seqd "?lsd_3 s=(10*n-3-wkimber[n-1])":
```

Theorem 12. *Kimberling's conjecture holds. The anti-Tribonacci sequence D_n and its complementary three sequences A_n, B_n, C_n satisfy Inequality 1.*

Proof. We need to verify that the sequences `seqa` up to `seqd` are indeed the (non)-anti-Tribonaccis. They need to be complementary, defined by the `mex`, and satisfy $A+B+C = D$. We first check that $A_n < B_n < C_n < A_{n+1}$, proving that A, B, C are disjoint:

```
eval test1 "?lsd_3 An,s,t,u,v ($seqa(n,s) & $seqb(n,t)
& $seqc(n,u) & $seqa(n+1,v)) => ((s<t)&(t<u)&(u<v))":
```

Next, we verify that the union of the four sequences is equal to $\mathbb{N}$:

```
eval test2 "?lsd_3 As (s>0) => Em ($seqa(m,s)|$seqb(m,s)|$seqc(m,s)|$seqd(m,s))":
```

In both cases `Walnut` says `TRUE`. We now verify that D is disjoint from $A \cup B \cup C$ to establish that the sequences are complementary:

```
eval test3 "?lsd_3 Em,n,s $seqd(n,s) & ($seqa(m,s) | $seqb(m,s) | $seqc(m,s))":
```

`Walnut` returns `FALSE` which confirms that the sequences are complementary. Finally, we check that $A + B + C = D$:

```
eval test4 "?lsd_3 An,s,t,u,v ($seqa(n,s) & $seqb(n,t) &
$seqc(n,u) & $seqd(n,v)) => v=s+t+u":
```

`Walnut` says `TRUE`. From $A + B + C = D$ we easily deduce that $D_n > C_n$ and therefore A, B, C are indeed defined by the `mex`, confirming the conjecture. □

Kimberling also conjectured that $A_{n+4} + A_n = A_{n+3} + A_{n+1}$, which is verified by:

```
eval kimconj "?lsd_3 An,r,s,t,u
($seqa(n,r) & $seqa(n+1,s) & $seqa(n+3,t) & $seqa(n+4,u)) => r+u=s+t":
```

As any clergyman will tell you, there is no need to stop at four. Indeed, Kimberling also considered the five complementary sequences A_n, B_n, C_n, D_n, E_n in which E_n are the *anti-Teranaccis* and the other sequences are the non-anti-Teranaccis (or missing numbers) defined by `mex`. Again, these sequences are complementary and the anti-Teranaccis are sums



Figure 6: The guessed `lsd_2`-automaton for the anti-Teranacci remainder.

of the missing numbers. The anti-Teranaccis form sequence A299409 where it is conjectured that

$$\begin{array}{rcrcl} 0 & \le & 17n-11-4A_n & \le & 4, \\ 0 & \le & 17n-\ 7-4B_n & \le & 4, \\ 0 & \le & 17n-\ 3-4C_n & \le & 3, \\ 0 & \le & 17n+\ 1-4D_n & \le & 3, \\ 0 & \le & 17n-\ 5-\ E_n & \le & 3. \end{array}$$

Again, we can guess automata for the remainder sequences, which we call `xx`, `yy`, `zz`, `vv`, `ww` this time, and define the sequences `A,B,C,D,E` from them. It turns out that there is a typo in the conjecture. The remainder of C is bounded by 4 instead of 3.

```
def seq4A "?lsd_2 s=(17*n-11-xxkimber[n-1])/4":
def seq4B "?lsd_2 s=(17*n-7-yykimber[n-1])/4":
def seq4C "?lsd_2 s=(17*n-3-zzkimber[n-1])/4":
def seq4D "?lsd_2 s=(17*n+1-vvkimber[n-1])/4":
def seq4E "?lsd_2 s=(17*n-5-wwkimber[n-1])":
```

We run the same four tests which again return `TRUE, TRUE, FALSE, TRUE`, as required.

```
eval test1 "?lsd_2 An,s,t,u,v,w ($seq4A(n,s) & $seq4B(n,t)
& $seq4C(n,u) & $seq4D(n,v) & $seq4A(n+1,w)) => ((s<t)&(t<u)&(u<v)&(v<w))":

eval test2 "?lsd_2 As (s>0) => Em ($seq4A(m,s) | $seq4B(m,s) | $seq4C(m,s) |
$seq4D(m,s) | $seq4E(m,s))":

eval test3 "?lsd_2 Em,n,s $seq4E(n,s) & ($seq4A(m,s) | $seq4B(m,s) |
$seq4C(m,s) | $seq4D(m,s))":
```

```
eval test4 "?lsd_2 An,s,t,u,v,w ($seq4A(n,s) & $seq4B(n,t) & $seq4C(n,u) &
$seq4D(n,v) & $seq4E(n,w)) => w=s+t+u+v":
```



Figure 7: The four guessed `lsd_2`-automata for the remainders of the non-anti-Teranaccis.

We define the anti-k-naccis as the sums of k consecutive missing numbers. There is an obvious pattern.

Conjecture 13 (The Clergyman's Conjecture)**.** Let X_n be the sequence of anti-k-naccis. The difference $X_n - (k^2+1)n$ is k-automatic.

5 Sumfreeness

Motivated by problems that were posed by Stephan [17] on the occasion of the 100,000-th entry in the OEIS, we consider *greedy 3-sumfree sequences* in this section. Consider an increasing sequence a of positive integers, defined by starting values a_1, a_2, a_3, extended by the rule that a_n, for $n > 3$ satisfies that a_n is the smallest number exceeding a_{n-1} that is not the sum of three different previous entries: so $a_n \neq a_i + a_j + a_k$ for $1 \leq i < j < k < n$. In his note [17] Stephan lists several conjectures (in Section 2.3.1) of the following kind. If $a = (1, 2, 3, \ldots)$ then $a_{n+6} - a_{n+5} = a_{n+1} - a_n$ for $n > 6$.

Queneau [13] defined an s-additive sequence with base $(a_1, a_2, ..., a_m)$ as the infinite sequence $a_1, a_2, a_3, \ldots$ with a_n, for $n > m$, equal to the least integer exceeding a_{n-1} with s

solutions of the form $a_n = a_i + a_j$ with $1 \le i < j \le n-1$. The special case $s = 0$ is essentially what we could call 2-sumfree sequences; these are also referred to in the literature as *sumfree* sequences (although sometimes the stricter condition $a_n \neq a_i + a_j$ with $1 \le i \le j \le n-1$ is then imposed). Apparently, it remains an open question if such sequences are always ultimately periodic, by which is meant that the sequence of first differences $(a_{n+1} - a_n)_{n \ge 1}$ would be ultimately periodic, see Finch [5].

We will use the notation $S_{x,y,z}$ for the greedy 3-sumfree sequence with starting values $x < y < z$. The cases $S_{1,2,3}$ and $S_{1,3,4}$ appear as A026471 and A026475 in the OEIS.

Since it was not clear to us how to verify conjectures like that of Stephan using `Walnut`, we first reworked them in a more explicit form that is amenable to direct verification, and then attempted to do this in `Walnut`. This led to families of conjectures of increasing generality (see Conjectures 14, 16 and 17 below) for (almost all) greedy 3-sumfree sequences starting with 1. The example for $(1, 2, 3, \ldots)$ reads as follows.

Conjecture 14. Let z be a positive integer. Then

$$z \in S_{1,2,3} \quad \Longleftrightarrow \quad z \in 1, 5, 13 \ \text{ or } \ z \bmod 23 \in \{2, 3, 4, 14, 15\}.$$

It is easy to verify by hand that $S_{1,2,3}$ starts as:

$$S_{1,2,3} = 1, 2, 3, 4, 5, 13, 14, 15, 25, 26, 27, 37, 38, 48, 49, 50, 60, 61, \ldots$$

and the conjecture would imply that from the seventh entry '14' on, the sequence modulo 23 is periodic with period 5.

With a few lines of `Walnut` code we can indeed verify this! First define residue classes modulo 23:

```
def res "Ek z=k*23+r":
```

which enables us to define the characteristic function for the sequence from the conjecture:

```
def seq123 "(z=1 | z=5 | z=13 | $res(2,z) | $res(3,z) | $res(4,z) |
$res(14,z) | $res(15,z))":
```

and we can now test the property that a positive integer z is in the sequence if and only if it is not the sum of 3 previous entries:

```
eval prop "Az z>0 => ($seq123(z) <=> z>0 & ~(E a, b, c a<b & b<c &
$seq123(a) & $seq123(b) & $seq123(c) & a+b+c=z))":
```

`Walnut` returns `TRUE` in little over a second: our conjecture is now a Theorem! We later noticed that according to a note (without reference) in A026471 this was also proved by Matthew Akeran.

We state (and proved by `Walnut`) this also for two more cases, $S_{1,3,4}$ and $S_{1,4,5}$.

Theorem 15. *For positive integers z:*

$$\begin{aligned}
z \in S_{1,2,3} &\iff z \in 1, 5, 13 \ \ \textit{or} \ \ z \bmod 23 \in \{2, 3, 4, 14, 15\};\\
z \in S_{1,3,4} &\iff z \in 1, 7, 19 \ \ \textit{or} \ \ z \bmod 33 \in \{3, 4, 5, 6, 20, 21, 22\};\\
z \in S_{1,4,5} &\iff z \in 1, 9, 25 \ \ \textit{or} \ \ z \bmod 43 \in \{4, 5, 6, 7, 8, 26, 27, 28, 29\}.
\end{aligned}$$

In fact we believe the following generalized conjecture to hold.

Conjecture 16. For every $g \geq 2$ the greedy 3-sumfree sequence $S_{1,g,g+1}$ is characterized by:

$$\begin{aligned}
z \in S_{1,g,g+1} \iff\ & z \in \{1, 2g+1, 6g+1\} \ \text{ or } \ z \bmod 10g+3 \in \\
& \{g, g+1, \ldots, 2g\} \cup \{6g+2, 6g+3, \ldots, 7g+1\}.
\end{aligned}$$

In particular, after the first $g+4$ entries the sequence modulo $10g+3$ is periodic with period $2g+1$.

Although such statement is not difficult to prove for given g, the parametrized statement is not so easily proved in `Walnut`. To be more precise, what one could attempt is to define the following:

```
def mod "Ek,w (((w>=g & w<=2*g) | (w>=6*g+2 & w<=7*g+1)) & z=k*(10*g+3)+w)":
def inG "z=1 | z=2*g+1 | z=6*g+1 | $mod(g,z)":
```

and then universally quantify (over g) this property:

```
eval prop "Az z>=g+4 => ($inG(z) <=> ~(E a, b, c a<b & b<c & a+b+c=z &
$inG(a) & $inG(b) & $inG(c) ))":
```

The problem here is that the definition of `mod` is not allowed in `Walnut` because of the multiplication by variables k and g:

```
the operator * cannot be applied to two variables
```

Whenever a numerical value is substituted for g in the definition of `mod` and `inG` all is fine. In fact we did this for all $g \in \{2, 3, \ldots 10\}$ successfully.

As a matter of fact we have firm computational evidence, from an implementation in Magma [3], for the following meta-conjecture.

Conjecture 17. Let $d \geq 2$. For every $g \geq d+1$ the greedy 3-sumfree sequence $S_{1,g,g+d}$ is characterized as follows:

$$\begin{aligned}
z \in S_{1,g,g+d} \iff\ & z \in \{1, g, 2g+d-1, 2g+d\} \ \text{ or } \ z \geq g+d \text{ and} \\
& z \bmod 5g+2d \in \{g+d-2, g+d-1, \ldots, 2g+d-2\}.
\end{aligned}$$

In particular, for $d \geq 2$ and every $g \geq d+1$ after the first $g+3$ entries (in a preperiod) the sequence $S_{1,g,g+d}$ modulo $5g+2d$ is periodic with period $g+1$.

Remarks 18. Note that Conjecture 16, although of the same form, is not a special case of the meta-conjecture, as the specific values (for modulus and (pre)period length) are irregular.

Also note that most, but not *all*, greedy 3-sumfree sequences are covered by the meta-conjecture: usually the cases of small values for g for given d are special in the sense that the indicated modular periodicity occurs for deviating values of modulus m and period length p. Below is one example.

In the regular cases of the meta-conjecture, it is also possible to state explicitly what the shape of the period (and preperiod) will be.

Example 19. For $d = 7$ and $g = 5$ the sequence $S_{1,5,12}$ becomes periodic modulo 321 with period length 32.

Here, by way of example, is a verification of the case $d = 4, g = 4$ of Conjecture 17. We first define the right residue classes modulo $5g + 2d = 28$ and then ask whether or not the 3-sumfree property holds for all entries in $S_{1,4,8}$ after $1, 4, 8$:

```
def isres "Eh, k h>= 6 & h<=10 & z=k*28+h":
def seq148 "(z=1 | z=4 | z=11 | z=12 | (z>=8 & $isres(z)))":
eval prop "Az z>8 => ($seq148(z) <=> ~(E a, b, c a<b & b<c
& $seq148(a) & $seq148(b) & $seq148(c) & a+b+c=z))":
```

which returns `TRUE` almost instantly.

Finally, it may be worthwhile to say something about the manner in which these conjectures were obtained. At first sight it is not clear at all that the definition of sumfree sequence leads to automatic sequences. Since this is important for the success of any attempt to invoke `Walnut`, it is useful to see how it arises naturally. But also, although not fitting in with the nature of this paper, it is useful for finding 'pen and paper' proofs for the claims, some of which we hope to present elsewhere.

The simple case is that it pays off to look at the sequence of first differences of these sumfree sequences. Consider again $S_{1,2,3}$. A straightforward computation leads to this initial segment:

$$S_{1,2,3} = 1, 2, 3, 4, 5, 13, 14, 15, 25, 26, 27, 37, 38, 48, 49, 50, 60, 61, 71, 72, 73, \ldots$$

and the corresponding sequence of differences:

$$D = 1, 1, 1, 1, 8, 1, 1, 10, 1, 1, 10, 1, 10, 1, 1, 10, 1, 10, 1, 1, \ldots.$$

Considering this it is not difficult to infer the conjectured periodicity.

6 Conclusion

We would like to thank several people that assisted us along the way. John Nicol and Markus Frohme ran the `co5` automaton for us on their upcoming new version of `Walnut`. Rob Burns

pointed out an omission in an earlier draft of this paper. Jeffrey Shallit helped us out with `Walnut` commands and gave a lecture in our mastermath course. The final draft of this paper was written during a visit of the workshop on Uniform Distribution of Sequences at the Erwin Schrödinger Institute in Vienna. We would like to thank the ESI staff for their hospitality.

References

[1] Boris Adamczewski and Jason Bell, On vanishing coefficients of algebraic power series over fields of positive characteristic, *Invent. Math.* **187** (2012), 343–393.

[2] Boris Adamczewski, Alin Bostan, and Xavier Caruso, A sharper multivariate Christol's theorem with applications to diagonals and Hadamard products. Arxiv preprint arXiv:2306.02640 [math.NT], 2023. Available at https://arxiv.org/abs/2306.02640.

[3] Wieb Bosma, John Cannon, and Catherine Playoust, The Magma algebra system I: The user language, *J. Symbolic Comput.* **24** (1997), 235–265.

[4] Aldo de Luca, A combinatorial property of the Fibonacci words, *Inf. Process. Lett.* **12** (1981), 193–195.

[5] Steven R. Finch, Are 0-additive sequences always regular?, *Amer. Math. Monthly* **99** (1992), 671–673.

[6] Nathan J. Fine, Binomial coefficients modulo a prime, *Amer. Math. Monthly* **54** (1947), 589–592.

[7] Lajos Hajdu, Robert Tijdeman, and Nóra Varga, On polynomials with only rational roots, *Mathematika* **69** (2023), 867–878.

[8] Sutasinee Kawsumarng, Tammatada Khemaratchatakumthorn, Passawan Noppakaew, and Prapanpong Pongsriiam, Sumsets associated with Wythoff sequences and Fibonacci numbers, *Period. Math. Hung.* **82** (2021), 98–113.

[9] Clark Kimberling and Peter J. C. Moses, Linear complementary equations and systems, *Fibonacci Quart.* **57** (2019), 97–111.

[10] Hamoon Mousavi, Automatic theorem proving in Walnut. Arxiv preprint arXiv: 1603.06017 [cs.FL], 2016. Available at https://arxiv.org/abs/1603.06017.

[11] John Nicol and Markus Frohme, Deconstructing subset construction—reducing while determinizing. Arxiv preprint arXiv:2505.10319 [cs.FL], 2025. Available at https://arxiv.org/abs/2505.10319.

[12] Phakhinkon Napp Phunphayap, Prapanpong Pongsriiam, and Jeffrey Shallit, Sumsets associated with Beatty sequences, *Discrete Math.* **345** (2022), 112810.

[13] Raymond Queneau, Sur les suites s-additives, *J. Comb. Theory Ser. A* **12** (1972), 31–71.

[14] Jeffrey Shallit, *The Logical Approach to Automatic Sequences: Exploring Combinatorics on Words with Walnut*, Vol. 482 of *London Math. Soc. Lecture Note Ser.*, Cambridge University Press, 2022.

[15] Jeffrey Shallit, Sumsets of Wythoff sequences, Fibonacci representation, and beyond, *Period. Math. Hung.* **84** (2022), 37–46.

[16] Jeffrey Shallit, Proving results about OEIS sequences with Walnut, In C. Dubois and M. Kerber, editors, *International Conference on Intelligent Computer Mathematics*, Vol. 14101 of *Lecture Notes in Artificial Intelligence*, pp. 270–282. Springer, 2023.

[17] Ralf Stephan, Prove or disprove 100 conjectures from the OEIS. Arxiv preprint arXiv: math/0409509v4 [math.CO], 2004. Available at https://arxiv.org/abs/math/0409509.

2020 *Mathematics Subject Classification*: Primary 11B83; Secondary 11R09, 68R15.
Keywords: integral polynomial, linear anti-recurrence, sumfree set, sum set, Wythoff number.

(Concerned with sequences A026471, A026475, A075326, A249031, A260311, A260317, A265389, A290409, A297464, A297465, A297466, and A299409.)

Received March 12 2025; revised versions received May 4 2025; July 16 2025. Published in *Journal of Integer Sequences*, July 19 2025.

Return to Journal of Integer Sequences home page.